

# An Executable Specification of Typing Rules for Extensible Records based on Row Polymorphism

Ki Yung Ahn

School of Computer Science and Engineering  
Nanyang Technological University, Singapore  
kyagr@gmail.com

Type inference is an application domain that is a natural fit for logic programming (LP). LP systems natively support unification, which serves as a basic building block of typical type inference algorithms. In particular, polymorphic type inference in the Hindley–Milner type system (HM) can be succinctly specified and executed in Prolog. In our previous work, we have demonstrated that more advanced features of parametric polymorphism beyond HM, such as type-constructor polymorphism and kind polymorphism, can be similarly specified in Prolog. Here, we demonstrate a specification for records, which is one of the most widely supported compound data structures in real-world programming languages, and discuss the advantages and limitations of Prolog as a specification language for type systems. Record types are specified as order-irrelevant collections of named fields mapped to their corresponding types. In addition, an open-ended collection is used to support row polymorphism for record types to be extensible.

## 1 Introduction

Static types are considered indispensable in areas of software development where safety and performance are critical. On the other hand, there has been a strong preference for dynamic languages in certain domains (e.g., web development) where they traditionally valued programmer productivity over safety and performance. However, changes in the software landscape during the last two decades are demanding higher standards of safety and efficiency in web and mobile applications. As a result, it has become a recent trend trying to develop static type systems for dynamic languages, some of which are being supported by major corporations and the original language developers themselves (e.g., Flow type checker [7], TypeScript [16], and Typed Lua [15]). In other words, static type systems are being re-invented in the real-world driven by the software industry demands [5]. Despite such demands, implementing type systems with rich modern features including type inference is still a challenge in practice (i.e., expensive and time consuming) due to the lack of automated tools and systematic methods. In contrast, lexers and parsers have been automatically generated from declarative grammar specifications ever since lex and yacc were introduced in the 70s.

Applications of logic programming (LP) for type inference has been investigated long before this recent trend of re-inventing static type systems in “the real-world”, at least from a decade ago (e.g., [17]). Some of the results from such investigations have been implemented in industry-strength compilers with advanced type systems such as the Glasgow Haskell Compiler [24]. Several other examples of LP-related applications for type systems are discussed in Section 6. However, there is not a widely used generic tool for deriving automated implementations of modern type systems, with polymorphic type inference from their declarative specifications. For syntax analysis, on the contrary, there are widely used tools (e.g., lex, yacc), that can generate lexers and parsers from their grammar specifications. In order to develop

generic tools for automating type system implementations for modern programming languages, we take an approach summarized by the following steps:

- (1) Identify features of type systems that are commonly available (or frequently wished for).
- (2) Formulate executable specifications of pedagogical type systems for minimalistic languages that support those features using a widely available LP system such as Prolog.
- (3) Examine the advantages and limitations of the LP system used the specifications.
- (4) Design and develop an LP-based tool for type system specification, strengthening the advantages and overcoming the limitations.
- (5) Apply the tool to describe full features of the advanced type systems of modern programming languages and examine further advantages and shortcomings for improvements.

We have gone through several iterations of steps 1, 2, and 3 over several features of parametric polymorphism, which are typically found in the type systems of functional languages, in our previous work [2]. In this work, we iterate through the same steps 1, 2, and 3 over record types, more specifically, extensible records based on row polymorphism [10]. Steps 4 and 5 are still left for future work.

Because row polymorphism extends type-constructor polymorphism, we review the specification for type-constructor polymorphism in Section 2. Then, we formulate a specification for a type system with extensible records in Section 3. Next, in Section 4, we summarize how to represent order-irrelevant structures using membership constraints, as in [21], in the context of applying such methods for record typing. Afterwards, we discuss related work in Section 6 and future work in Section 7.

Let us review a specification for the Simply-Typed Lambda Calculus (STLC) in Prolog (see Fig. 1) as a precursor to reviewing the specifications of polymorphic type systems in Section 2. Observe that the specification in Fig. 1 are structurally similar to the well-known typing rules of STLC shown below:<sup>1</sup>

$$\begin{array}{c} \text{(VAR)} \frac{x : A \in \Gamma}{\Gamma \vdash x : A} \quad \text{(LAM)} \frac{x : A, \Gamma \vdash e : B}{\Gamma \vdash \lambda x. e : A \rightarrow B} \quad \text{(APP)} \frac{\Gamma \vdash e_1 : A \rightarrow B \quad \Gamma \vdash e_2 : A}{\Gamma \vdash e_1 e_2 : B} \end{array}$$

More specifically, lines 7-9 of Fig. 1 correspond to the rules (VAR), (LAM), and (APP), respectively. The conclusion of each typing rule of STLC corresponds to the left-hand side of :- in each line of the definition for the "type" predicate and the premiss of each typing rule corresponds to the right-hand side. The test for variable binding in the typing context  $x : A \in \Gamma$  in the (VAR) rule is implemented as the Prolog predicate `first`, which searches for first occurrence of a key-value pair in the list. We assume that readers are familiar with the extra-logical built-in Prolog predicate `\==` in the definition of `first`.

Because this specification is *relational*, rather than functional, it describes both type checking and type inference without duplication. In addition, the specification is *executable* on a machine using a Prolog implementation. All of our specifications (Figs. 1, 2, 3, and 4) in this article have been tested on SWI Prolog version 7.3.2. Here are some queries on the STLC specification:

```
?- type([], lam(x, var(x)), A->A).
true .

?- type([], lam(x, var(x)), T).
T = (_G1622->_G1622) .

?- type([], E, A->A).
E = lam(_G16633, var(_G16633)) .
```

<sup>1</sup> It is a more standard notation to append the variable binding to the right, that is  $\Gamma, x : A$ , in the (LAM) rule, but we append it to the left, that is,  $x : A, \Gamma$  to emphasize the structural similarity with the Prolog specification in Fig. 1.

```

1 :- set_prolog_flag(occurs_check,true).
2 :- op(500,xfy,$).
3
4 first(K:V,[K1:V1|_]) :- K = K1, V = V1.
5 first(K:V,[K1:_|Xs]) :- K\==K1, first(K:V, Xs).
6
7 type(C,var(X),      A) :- first(X:A,C).
8 type(C,lam(X,E),A->B) :- type([X:A|C], E, B).
9 type(C,E1 $ E2,     B) :- type(C,E1,A->B), type(C,E2,A).

```

Figure 1: Specification of the Simply-Typed Lambda Calculus (STLC) in Prolog

```

7 type(C,var(X),      A) :- first(X:S,C), inst_ty(S,A).
8 type(C,lam(X,E),A->B) :- type([X:mono(A)|C],E,B).
9 type(C,E1 $ E2,     B) :- type(C,E1,A->B), type(C,E2,A).
10 type(C,let(X=E0,E),T) :- type(C,E0,A),
11                          type([X:poly(C,A)|C],E,T).
12
13 inst_ty(poly(C,T),T1) :- copy_term(t(C,T),t(C,T1)).
14 inst_ty(mono(T), T ).

```

Figure 2: Specification of the Hindley–Milner type system (HM) in Prolog

```

7 kind(KC,var(Z), K) :- first(Z:K,KC).
8 kind(KC,F $ G, K2) :- kind(KC,F,K1->K2), kind(KC,G,K1).
9 kind(KC,A -> B, o) :- kind(KC,A,o), kind(KC,B,o).
10
11 type(KC,C,var(X),      A) --> { first(X:S,C) }, inst_ty(KC,S,A).
12 type(KC,C,lam(X,E),A->B) --> type(KC,[X:mono(A)|C],E,B),
13                               [ kind(KC,A->B,o) ].
14 type(KC,C,E1 $ E2,     B) --> type(KC,C,E1,A->B), type(KC,C,E2,A).
15 type(KC,C,let(X=E0,E),T) --> type(KC,C,E0,A),
16                               type(KC,[X:poly(C,A)|C],E,T).
17
18 inst_ty(KC,poly(C,T),T2) --> { copy_term(t(C,T),t(C,T1)),
19                               free_variables(T,Xs),
20                               free_variables(T1,Xs1) },
21                               samekinds(KC,Xs,Xs1), { T1=T2 }.
22 inst_ty(_, mono(T), T) --> [].
23
24 samekinds(KC,[X|Xs],[Y|Ys]) --> { X\==Y },
25                               [ kind(KC,X,K), kind(KC,Y,K) ],
26                               samekinds(KC,Xs,Ys).
27 samekinds(KC,[X|Xs],[X|Ys]) --> [], samekinds(KC,Xs,Ys).
28 samekinds(_ ,[],      []) --> [].

```

Figure 3: Specification of HM extended with type-constructor polymorphism in Prolog

The specification of STLC (Fig. 1) can be executed for both *type checking* and *type inference* as needed. In addition, it can also be executed for *type inhabitation*, although not the main focus of this work. The other specifications (Figs. 2, 3, and 4) in this work are developed in similar manner in order to keep the advantages of executable relational specifications based on LP, as demonstrated by the queries on STLC.

Our contributions are:

- development of an executable relational specification of typing rules for extensible records by incrementally extending the type system to support simple types (Fig. 1), type and type-constructor polymorphisms (Section 2), and row polymorphism (Section 3),
- implementation of type inference for extensible records using open-ended order-irrelevant structures in logic programming based on membership constraints (Section 4), and
- analysis of the benefits and limitations of Prolog for type system specification from our experience of developing specifications for extensible records (Section 5).

## 2 Specifications of polymorphic type systems using Prolog

Our previous work [2] demonstrates that LP is suitable for declarative and executable specifications of polymorphic type systems including advanced features such as type-constructor polymorphism (a.k.a. higher-kinded polymorphism) and kind polymorphism, in addition to the type polymorphism in HM. In this section, we review the specifications for polymorphisms over types (Section 2.1) and type-constructors (Section 2.2). Later on, in Section 3, we extend upon the specification for type-constructor polymorphism to specify row polymorphism.

Lines 1-6 in Figs. 2 and 3 are omitted because they are exactly the same as the lines 1-6 in Fig. 1. The first line enables occurs check in SWI Prolog and the second line declares an infix syntax for representing application terms. The lines 4-5 defines a first-occurrence lookup predicate for key–value pairs, which is used for handling both type bindings and kind bindings.

### 2.1 Type polymorphism in the Hindley–Milner type system (HM)

The Hindley–Milner type system (HM) is a conservative extension of STLC. That is, all true typing judgements in STLC are also true in HM. We can observe this by comparing the STLC specification in Fig. 1 and the HM specification in Fig. 2. We can convert any true typing judgement of STLC represented by the `type`-relation into true typing judgement in HM by replacing every type binding  $X:A$  with a monomorphic binding  $X:\text{mono}(A)$  because `inst_ty` simply instantiates `mono(A)` into  $A$ .

There is another kind of type binding in HM in the form of  $X:\text{poly}(C,T)$ , which appears in lines 11 and 13 of Fig. 2. This is a polymorphic binding that gives the additional power to HM compared to STLC. A representative example is the polymorphic identity function. Consider the following type-inference query:

```
?- type([], let(id=lam(x,var(x)), var(id) $ var(id)), T).
T = (_G1703->_G1703) .
```

A `let`-term in HM that was not part of the STLC term syntax, which introduces polymorphic bindings. For instance, in the query above, a polymorphic binding for `id` is introduced into the typing context when inferring the type for the body of the `let`-term according to the lines 10-11 in Fig. 2. Assuming the type of `lam(x, var(x))` was inferred as  $A \rightarrow A$  by line 8, the type inference for the body `var(id)$var(id)` by line 9 corresponds to the following query:

```
?- type([id:poly([],A->A)], var(id) $ var(id), T).
T = (_G1697->_G1697) .
```

Each occurrence of `var(id)` is inferred a type with freshly instantiated using `copy_term`.<sup>2</sup> Say it was inferred as  $A1 \rightarrow A1$  for the former and  $A2 \rightarrow A2$  for the latter. Then,  $A1$  unifies with  $A2 \rightarrow A2$  because the latter should match with the argument type of the former function type. Thus, the type of `verb(id)$verb(id)` is inferred as  $A2 \rightarrow A2$  (the variable `_G1697` above corresponds to  $A2$ ).

In contrast to the `let`-term, bindings introduced by the lambda term are monomorphic. For example, if we use a lambda term instead of the `let`-term in the previous example, type inference fails:

```
?- type([], lam(id,var(id)$var(id)) $ lam(x,var(x)), T).
false.
```

More specifically, assuming that the type of `lam(x,var(x))` was inferred as  $A \rightarrow A$ , the query that attempts to infer the body of the lambda term, `var(id)$var(id)`, fails due to the occurs check:<sup>3</sup>

```
?- type([id:mono(A->A)], var(id) $ var(id), T).
false.
```

## 2.2 Extending HM with type-constructor polymorphism

**Type constructors** are type-level objects that take arguments to produce types. Type constructors are widely used in functional languages with polymorphic type systems for defining parametrized types. For example, consider the following datatype definitions in Haskell:

```
data Pair t = Pair (t,t)          data List a = Nil | Cons a (List a)
```

`Pair` and `List` are type constructors that take one type argument to produce a type. For instance, `Pair Int` is a type for integer pairs and `List Char` is a type for character lists. More generally, type constructors can take multiple arguments<sup>4</sup> and those arguments can be type constructors as well as types. For instance, the `Tree` datatype defined below takes a type constructor as its first argument (`c`) and a type as its second argument (`a`):

```
data Tree c a = Tip | Node a (c (Tree c a))
```

We know that `a` is a type because it appears alone as the first argument of `Node` in the definition of `Tree` and that `c` is a type constructor because it takes a type argument (`Tree c a`) to produce the second argument (`c (Tree c a)`).

Type constructors are categorized by *kinds* just as terms are categorized by types. All types are of kind `*`. For instance `Int :: *` and `List Char :: *`. Type constructors that take arguments have arrow kinds just as functions that take other terms as arguments have arrow types. For instance, `List :: * -> *` because it takes one type argument of kind `*` to produce types of kind `*` and `Tree :: * -> (* -> *) -> *` because it takes two arguments of kind `*` and `(* -> *)`.

The rules for categorizing types by their kinds are called *kinding rules*. A type system that supports type constructors includes kinding rules in addition to the typing rules that categorize terms by their types. Lines 7-9 in Fig. 3 specify such kinding rules in Prolog where the kind `*` is denoted as `o` because `*`

<sup>2</sup> The predicate `copy_term` is an extra-logical built-in predicate that copies the first argument with each of its uninstantiated variables made fresh and then unifies the copied term with the second argument. For example, querying `copy_term(poly([A],A->B), poly([A],T))` first makes a copy `poly([A1],A1->B1)`, where  $A1$  and  $B1$  are fresh, and then unifies with the second argument, resulting in  $T = A \rightarrow B1$  because  $A$  unifies with  $A1$ .

<sup>3</sup> The occurs check is enabled in line 3, which is omitted in Fig. 2 because it is the same as in Fig. 1.

<sup>4</sup> Types can also be considered as nullary type constructors that take zero arguments.

is predefined as an infix operator in Prolog. The kind predicate in Fig. 3 has virtually the same structure as the type predicate in Fig. 1. That is, the kinding rules in Fig. 3 can be considered as a replication of STLC on the type level, except for lambda terms that introduce local variable bindings. The kind variables bound in the *kinding context* (KC) in the kind predicate, as apposed to the typing context (C) in the type predicate, are introduced globally by the type instantiation rule `inst_ty`, more specifically, by `samekinds`. Because the typing rule calls upon kinding rules (e.g., line 13 of Fig. 3), the typing rule should also pass around the kinding context (KC) as an extra argument in addition to the typing context (C) throughout lines 11-16 of Fig. 3.

**Type-constructor polymorphism** supports polymorphic types involving not only type variables, which instantiate to types (i.e., nullary type constructors), but also type-constructor variables, which instantiate to non-nullary type constructors. In fact, we have already introduced an example that uses type-constructor polymorphism in its definition. Note that `c` in the definition `Tree` is a type-constructor variable of kind `*->*`, which can be instantiated to a type constructor that describes the branching structure of the tree. For example, we can define types for binary trees (every node has two branches) and rose trees (each node can have arbitrary number of branches) by instantiating the type constructor variable `c :: *->*` in the type-constructor polymorphic datatype `Tree` as follows:

```
type BinTree a = Tree Pair a
type RoseTree a = Tree List a
```

Both `BinTree` and `RoseTree` shares the data constructors of `Tree`, which are `Tip` and `Node`. These two data constructors can be used for building any specific tree instantiated from `Tree` because their types are polymorphic over `c` and `a`. For instance, `c` can be instantiated to `Pair`, `List`, `Tree Pair`, or `Tree List`, and, `a` can be instantiated to `Bool`, `List Char`, or `Tree Pair Int`. However, it is clearly wrong to instantiate `c` to `Int` or `a` to `List` because the variable and its instantiations are of different kinds. The `samekind` rule in lines 27-28 of Fig. 3 implements this sanity check of instantiating type constructor variables to type constructors of the same kind by generating a list of kind predicates, which must to be satisfied. The type instantiation rule `inst_ty` in lines 18-22 of Fig. 3 extends the `inst_ty` predicate of HM in lines 13-14 implemented by `copy_term` to to utilize the `samekind` rule.

**Two-staged execution of typing rules and kinding rules** using Definite Clause Grammars (DCGs) is another noticeable change from the HM specification. The long arrow syntax using `-->` defines DCG rules whereas the standard clause syntax using `:-` defines ordinary Prolog predicates. DCG rules invoke DCG rules in their definition by default (e.g., line 14 of Fig. 3) but they can also invoke ordinary predicates inside the curly braces (e.g., `{ first(X:S,C) }`) in line 11 of Fig. 3. What distinguishes a DCG from an ordinary predicate is that it produces a list as a result of its invocation. The items to be collected into the resulting list are put inside square brackets (e.g., lines 13 and 25 of Fig. 3). We have exploited the DCG rules to execute typing rules first while delaying the execution of the kinding rules until all unifications over type constructors have been resolved. Further details including how to invoke the two-staged execution of the specifications and why we execute the specifications in two stages are to be discussed in Sections 3 and 5.

### 3 A specification of extensible records based on row polymorphism

The key aspects of the gradually extended type systems from supporting type polymorphism in Fig. 2, type-constructor polymorphism in Fig. 3, and row polymorphism [10] in Fig. 4 are highlighted in their

```

7  kind(KC,var(X),    K1) :- first(X:K,KC).
8  kind(KC,F $ G,    K2) :- K2\==row, kind(KC,F,K1->K2),
9                               K1\==row, kind(KC,G,K1).
10 kind(KC,A -> B,    o) :- kind(KC,A,o), kind(KC,B,o).
11 kind(KC,{R},      o) :- kind(KC,R,row).
12 kind(KC,[],       row).
13 kind(KC,[X:T|R],  row) :- kind(KC,T,o), kind(KC,R,row).
14
15 type(KC,C,var(X),  A) --> { first(X:S,C) }, inst_ty(KC,S,A).
16 type(KC,C,lam(X,E),A->B) --> type(KC,[X:mono(A)|C],E,B),
17                               [ kind(KC,A->B,o) ].
18 type(KC,C,X $ Y,  B) --> type(KC,C,X,A->B), type(KC,C,Y,A1),
19                               !, { eqty(A,A1) }. % note the cut !
20 type(KC,C,let(X=EO,E),T) --> type(KC,C,EO,A),
21                               type(KC,[X:poly(C,A)|C],E,T).
22 type(KC,C,{XEs},  {R}) --> { zip_with('=',Xs,Es,XEs) },
23                               type_many(KC,C,Es,Ts),
24                               { zip_with(':',Xs,Ts,R) }.
25 type(KC,C,sel(L,X), T) --> { first(X:T,R) }, type(KC,C,L,{R}).
26
27 inst_ty(KC,poly(C,T),T2) --> { copy_term(t(C,T),t(C,T1)),
28                               free_variables(T,Xs),
29                               free_variables(T1,Xs1) },
30                               samekinds(KC,Xs,Xs1), { T1=T2 }.
31 inst_ty(KC,mono(T),  T) --> [].
32
33 samekinds(KC,[],    []) --> [].
34 samekinds(KC,[X|Xs],[Y|Ys]) --> { X\==Y },
35                               [ kind(KC,X,K), kind(KC,Y,K) ],
36                               samekinds(KC,Xs,Ys).
37 samekinds(KC,[X|Xs],[X|Ys]) --> [], samekinds(KC,Xs,Ys).
38
39 zip_with(F,[],    [],    []).
40 zip_with(F,[X|Xs],[Y|Ys],[FXY|Ps]) :- FXY=..[F,X,Y],
41                                       zip_with(F,Xs,Ys,Ps).
42
43 type_many(KC,C,[],    []) --> [].
44 type_many(KC,C,[E|Es],[T|Ts]) --> type(KC,C,E,T), type_many(KC,C,Es,Ts).

```

Figure 4: Specification of extensible records based on row polymorphism in Prolog

```

46  variablize(var(X)) :- gensym(t,X).
47
48  infer_type(KC,C,E,T) :-
49    phrase( type(KC,C,E,T), Gs0 ),
50    copy_term(Gs0,Gs),
51    (bagof(Ty,X^Y^member(kind(X,Ty,Y),Gs),Tys); Tys=[]),
52    free_variables(Tys,Xs),
53    maplist(variablize,Xs), % replace free tyvar to var(t)
54    maplist(call,Gs). % run all goals in Gs
55
56  ctx0([ 'Nat':mono(o)
57        , 'List':mono(o->o)
58        , 'Pair':mono(o->o->o)
59        | _
60        ],
61        [ 'Zero':mono(Nat)
62        , 'Succ':mono(Nat -> Nat)
63        , 'Nil' :poly([], List$A)
64        , 'Cons':poly([], A->((List$A)->(List$A)))
65        , 'Pair':poly([], A0->B0->Pair$A0$B0)
66        ])
67  :- Nat = var('Nat'), List = var('List'), Pair=var('Pair').
68
69  run(N,T) :- ctx0(KC,C),
70    Zero = var('Zero'), Succ = var('Succ'),
71    Cons = var('Cons'), Nil = var('Nil'),
72    Pair = var('Pair'),
73    E0 = let(id=lam(x,var(x)),var(id)$var(id)), % A->A
74    E1 = lam(y,let(x=lam(z,var(y)),var(x)$var(x))), % A->A
75    E2 = {[z=lam(x,var(x))]}, % {[z:A->A]}
76    E3 = lam(r,sel(var(r),x)), % {[x:A |R]} -> A
77    %E4: {[y:A,x:B |R]} -> Pair$A$B
78    E4 = lam(r,Pair$sel(var(r),y)$sel(var(r),x)),
79    %E5: {[x:(A->A),y:(B->B)]}
80    E5 = {[x=lam(x,var(x)),y=lam(x,var(x))]},
81    E6 = E4 $ E5, % Pair $ B->B $ A->A
82    %E7: {[y:A |R]} -> Pair $ A $ {[y:A | R]}
83    E7 = lam(r,Pair$sel(var(r),y)$var(r)),
84    E8 = E7 $ {[y=lam(x,var(x))]}, % Pair $ B->B $ {[y:(B->B)| R]}
85    E9 = E7 $ {[]}, % should not type check
86    E10 = {[]},
87    Es = [E0,E1,E2,E3,E4,E5,E6,E7,E8,E9,E10],
88    nth0(N,Es,E), infer_type(KC,C,E,T).

```

Figure 5: Running type inference for example terms with records against the specification in Fig. 4



kind systems. The kind system of HM is so trivial that it is typically omitted from the specification; we can think of HM as having the only one kind  $*$ , which we represent as  $o$  in Prolog. To support type-constructor polymorphism (Fig. 3), the kind syntax should be enriched as the following BNF grammar:  $\kappa ::= * \mid \kappa_1 \rightarrow \kappa_2$ . The kind system including row polymorphism needs an additional kind `row`, that is,  $\kappa ::= * \mid \text{row} \mid \kappa_1 \rightarrow \kappa_2$ . Instances of `row` are disjoint from types and type constructors, which are instances of  $*$  and  $\kappa_1 \rightarrow \kappa_2$ . An instance of `row` describes a collection of field names paired with their types, where each name is unique within the collection. In other words, it is a map from names to types.

Lines 7-10 of Fig. 4 are almost identical to the lines 7-9 of Fig. 3 except for `K1\==row` and `K2\==row`, which prevent hopeless instantiations of `K1` and `K2` because they should be either  $*$  or arrow kinds.

Lines 12-13 describes how to form an instance of `row`. An instance of `row` is either an empty collection `[]` or an existing row instance `R` extended with a name–type pair `X:T`. For simplicity, we will say “a row” where we mean “a type-level object of kind `row`” or “an instance of `row`”. Line 11 provides a way to construct a record type from a row. We use curly braces to give records and their types a clear syntactic distinction from the other language constructs. Curly braces around name–type pairs (i.e., rows) make record types and curly braces around name–term pairs make records. For example, the record `{[x=1,y=2]}` has type `{[x:int,y:int]}` provided that 1 and 2 are of type `int`.

Note that `{[x=1,y=2]}` can also have type `{[y:int,x:int]}`. The `eqty` predicate used in the typing rules for application terms (lines 18-19) is a generalized equality over types that handles this order-irrelevance of records and also the row-polymorphic nature of the record typing. Assuming that Prolog generates most general instances of `A` and `A1` first, we can use `cut(!)` to prevent backtracking when `eqty(A,A1)` fails; if it failed to equate the most general instances there is no hope for more specific ones. This use of `cut` in line 19 is necessary for the termination of type inference in case it fails for ill-typed terms involving record types. For example, the type inference for `E9` in Fig. 5 does not terminate without this `cut`; it hopelessly backtracks by populating the record type argument (`A`) for the function type of `E7` with one more row element each time, which could never be unified with `A1={[]}`. Details on the implementation of `eqty` will be discussed in Section 4.

Lines 22-24 specifies the typing rules for records. For example, the record `{[x=1,y=2]}` has type `{[x:int,y:int]}`. Line 25 specifies the typing rule for record field selection. For example, the result of evaluating `sel({x=2,y=3},x)` is going to be 2. Thus, the type of `sel(L,X)` must be the type of the field `X` provided that the record `L` has a field with name `X`; `first(X:T,R)` specifies that there should exist a row `R`, which contains a field with name `X` and its type `T`; `type(KC,C,L,{R})` specifies that the record `L` should indeed have the record type `{R}`.

The `infer_type` rule in Fig. 5 executes the specification in two stages. The Prolog code in Fig. 5 follows the Prolog specification in Fig. 4. Recall that the typing rules in Figs. 3 and 4 are defined as DCG rules, which produce delayed kinding assertions when executed. For type checking or inference, we execute the DCG rule `type` using `phrase` (see line 44 of Fig. 5) and then execute the kind predicates collected in the resulting list `Gs0`. More precisely, we execute them from `Gs`, a freshened copy of `Gs0`. Before we execute the kind assertions using `maplist` in line 55, we instantiate the type variables in them as concrete variables of the form `var(x)`, where `x` is a fresh atomic name, by using `variablize` in lines 52-54. The result of running type inference for each example is provided in the Prolog comment near the example term.

```

1 % setmemb.pl
2
3 memb(X,[X|_]).
4 memb(X,[Y|L]) :- X \== Y, memb(X,L).
5
6 % unify finite sets (i.e. set with [] at the end)
7 unify_set(A,B) :- subset_of(A,B), subset_of(B,A).
8
9 subset_of([],_).
10 subset_of([X|R],L) :- memb(X,L), subset_of(R,L).
11
12 % finite set union
13 union([],B,B).
14 union(A,[],A).
15 union([X|Xs],B,C) :- memb(X,B), !, union(Xs,B,C).
16 union([X|Xs],B,[X|C]) :- union(Xs,B,C).
17
18 % finite set minus
19 setminus(A,[],A).
20 setminus([],_,[]).
21 setminus([X|Xs],B,C) :- memb(X,B), !, setminus(Xs,B,C).

```

Figure 6: Operations on sets including unification based on membership constraints

```

1 % unify open ended sets with possibly uninstantiated variable tail
2 unify_oeset(A,B) :- ( var(A); var(B) ), !, A=B.
3 unify_oeset(A,B) :- split_heads(A,Xs-T1), make_set(Xs,S1),
4                     split_heads(B,Ys-T2), make_set(Ys,S2),
5                     unify_oe_set(S1-T1, S2-T2).
6
7 make_set(L,S) :- setof(X,memb(X,L),S). % remove duplicates
8 make_set([],[]).
9
10 split_heads([],[]-[]).
11 split_heads([X|T],[X]-T) :- var(T), !, true.
12 split_heads([X|Xs],[X|Hs]-T) :- split_heads(Xs,Hs-T).
13
14 % helper function for unify_oeset (almost no duplicate answers)
15 unify_oe_set(Xs-T1,Ys-T2) :- T1==[], T2==[], unify_set(Xs,Ys).
16 unify_oe_set(Xs-T1,Ys-T2) :- T1==[], subset_of(Ys,Xs),
17                               setminus(Xs,Ys,T2).
18 unify_oe_set(Xs-T1,Ys-T2) :- T2==[], subset_of(Xs,Ys),
19                               setminus(Ys,Xs,T1).
20 unify_oe_set(Xs-T1,Ys-T2) :- setminus(Ys,Xs,L1), append(L1,T,T1),
21                               setminus(Xs,Ys,L2), append(L2,T,T2).
22
23 %% ?- unify_oeset([1,3,5,9|T1],[2,3,5,6|T2]).
24 %% T1 = [2, 6|_G481], T2 = [1, 9|_G481].

```

Figure 7: Unification over open-ended sets using the membership constraints for finite sets in Fig. 6

## 4 Order-irrelevant structures in logic programming

We summarize the work [21] on representing sets using membership constraints by providing an executable Prolog code, which were not provided in Stolzenburg’s paper, in Section 4.1. Then, we discuss it in the context of record typing in Section 4.2.

### 4.1 Unification over sets based on membership constraints

Consider the problem of set equivalence. For example,  $\{x, y\}$ ,  $\{y, x\}$ ,  $\{x, y, x\}$ , and  $\{x, y, y\}$  should all be considered equivalent.<sup>5</sup> In logic programming, we can also ask the question of how to unify two representation of sets. For example, we expect that the unification of  $\{X, y\}$  and  $\{Y, x\}$  to be  $X=x$ ,  $Y=y$ . We implemented Stolzenburg’s work as a Prolog program (`setmemb.pl`) as shown in Fig. 6. The use of extra-logical built-in inequalities (`\==`) and cuts (!) reduces unnecessary equivalent solutions.

More generally, we can think of unification between possibly open-ended sets where the tail (or the rest) of the set is a variable. That is, their size may not be able to be determined before unification. For example, consider unifying a set containing at least  $x$  denoted by  $\{x|S1\}$  and a set containing at least  $y$  denoted by  $\{y|S2\}$ . We intuitively know that the most general unification for the query `?- {x|S1} = {y|S2}`. is  $S1 = \{x|S\}$ ,  $S2 = \{y|S\}$  where the substitution for  $S1$  and  $S2$  shares the same tail  $S$ . This has also been explored by Frieder Stolzenburg (1996) [21].

We implemented the description in his work in Prolog (`setunify.pl`) as shown in Fig. 7. The basic idea is to test whether the sets being unified are open-ended and treat them accordingly in each case:

- When both are closed-ended (line 25 of Fig. 7), call the predicate for closed-ended set unification (`unify_set`) from Fig. 6.
- When one of them is open-ended and the other is closed-ended (lines 26-29 of Fig. 7), instantiate the tail of the open-ended set with the difference set of the closed-ended set minus the leading elements of the open-ended set.
- When both are open-ended (lines 30-31 of Fig. 7), instantiate the tails with the difference sets of each others’ leading elements but keep them both open-ended by a common tail variable  $T$ .
- It is also possible that both sides of the set unification may be completely uninstantiated variables. Such a case is handled in lines 11 by the driver predicate `unify_oeset`, which calls the `unify_oe_set` to do the main work.

### 4.2 Type equality for records using nested open-ended map unification

Open-ended set unification are not widely supported (i.e., either a built-in native unification operation or shipped as a standard library) in Prolog implementations because nesting of open-ended sets adds complications to implementing a unification algorithm. Prior work on this subject (e.g., [21]) assumes non-nested sets. That is, the elements of sets are treated as ordinary structures rather than taking into consideration the possibility of those elements being a set.

For record typing we need nested open-ended structures because records can have elements which are records. Fortunately for extensible records, it is easier to handle nested structures because the field names of records are always atomic names (e.g.,  $x$  and  $y$  in  $\{x: \text{int}, y: \text{bool}\}$ ). In other words, nested open-ended unification is more manageable for maps than it is for sets because tagged-sums are easier to handle than (untagged) unions.

---

<sup>5</sup>Some Prolog implementations use curly braces for tuples. Here, we are not using any pair operations of such, just to clarify.

```

1 eqty(A1,A2) :- (var(A1); var(A2)), !, A1=A2.
2 eqty({R1},{R2}) :- !, unify_oemap(R1,R2). % like a permutation
3 eqty(A1->B1,A2->B2) :- !, eqty(A2,A1), !, eqty(B1,B2).
4 eqty(A,A).
5
6 % unify finite maps
7 unify_map(A,B) :- submap_of(A,B), submap_of(B,A).
8
9 submap_of([],_).
10 submap_of([X:V|R],M) :- first(X:V1,M), eqty(V,V1), submap_of(R,M).
11
12 % finite map minus
13 mapminus(A,[],A).
14 mapminus([],_,[]).
15 mapminus([X:V|Ps],B,C) :- first(X:V1,B),
16                             !, eqty(V,V1) -> mapminus(Ps,B,C).
17 mapminus([X:V|Ps],B,[X:V|C]) :- mapminus(Ps,B,C).
18
19 % unify open ended maps with possibly uninstantiated variable tail
20 unify_oemap(A,B) :- ( var(A); var(B) ), !, A=B.
21 unify_oemap(A,B) :- split_heads(A,Xs-T1), make_map(Xs,M1),
22                       split_heads(B,Ys-T2), make_map(Ys,M2),
23                       unify_oe_map(M1-T1, M2-T2).
24
25 make_map(L,M) :- setof(X:V,first(X:V,L),M). % remove duplicates
26 make_map([],[]).
27
28 split_heads([],[]-[]).
29 split_heads([X:V|T],[X:V]-T) :- var(T), !, true.
30 split_heads([X:V|Ps],[X:V|Hs]-T) :- split_heads(Ps,Hs-T).
31
32 % helper function for unify_oemap
33 unify_oe_map(Xs-T1,Ys-T2) :- T1==[], T2==[], unify_map(Xs,Ys).
34 unify_oe_map(Xs-T1,Ys-T2) :- T1==[],
35                               submap_of(Ys,Xs), mapminus(Xs,Ys,T2).
36 unify_oe_map(Xs-T1,Ys-T2) :- T2==[],
37                               submap_of(Xs,Ys), mapminus(Ys,Xs,T1).
38 unify_oe_map(Xs-T1,Ys-T2) :- mapminus(Ys,Xs,L1), append(L1,T,T1),
39                               mapminus(Xs,Ys,L2), append(L2,T,T2).
40
41 %% ?- unify_oemap([z:string,y:bool|M1],[y:T,x:int|M2]).
42 %% M1 = [x:int|_G1426], T = bool, M2 = [z:string|_G1426].

```

Figure 8: Type equality for records using open-ended map unification.

In Fig. 4, we define the type equality including record types (`eqty` in Fig. 8), which is defined in terms of nested open-ended map unification (`unify_oemap`). Unification of open-ended maps are similar in structure to the open-ended sets discussed previously. One difference from the open-ended set unification in Fig 7 is that definitions in Fig. 8 can handle nested record types properly because `eqty` and `unify_oemap` are mutually recursive definitions (`unify_oemap` calls `mapminus`, which calls `eqty`).

## 5 Limitations

In the introduction, we have already mentioned the advantages of executable relational specifications. Here, we briefly discuss the limitations of our specification of the type system for extensible records.

**Multi-staged execution with multiple sets of logic variables:** The kinding and typing rules in our specifications are not only executed in two different stages but also distinguish the set of logic variables in each stage. The type variables can be instantiated into a more specific type during the type inference stage. However, in the kind inference stage after type inference, the type variables are no longer considered as logic variables but atomic constants that cannot be unified with anything else than themselves. In our Prolog specification (Fig. 5) we instantiate all the type variables into fresh atomic constants using `gensym` built-in Prolog predicate before kind inference. Having a more systematic way of managing multi-staged execution of logic programs with different set of logic variables for each stage will be especially helpful for specifying type systems with more than just two stages (e.g., [25, 1]).

**Explicit use of cuts and type equality in the specification:** The specifications for polymorphisms over types, type-constructors, and kinds did not require explicit use of cuts (!) in the predicates specifying the typing rules; cuts were contained inside helper predicates such as `first`. However, the specification for extensible records in Fig. 4 has a cut in the typing rule for applications. In addition, it is also more explicit about where to use the type equality (`eqty`) beyond the native structural unification of Prolog. We think that explicit use of more advanced type equality may be a reasonable compromise; in comparison to the specifications involving qualified type systems by Jones [13], our specification for extensible records is less declarative than the syntax-directed typing rules for qualified types but it is still more declarative than the inference-rule style of specification of the W-algorithm for qualified types. However, the use of cut just before `eqty` in the application rule (line 18 in Fig. 4) is not so desirable as a declarative specification.

**Error reporting:** All practical type system implementations must report error messages. Currently, our specification as a Prolog program outputs `false` by default when the type inference fails without any useful information of why it failed. We can of course print error messages by manually adding Prolog primitives within the specification as needed but unprincipled addition of error handling functions may obscure the specification. We hope to find a more systematic method of handling error messages.

## 6 Related work

Earlier in the 80's, `Typol`, which compiles into Prolog, was used for the executable specification of type systems [6]; the examples in the paper [6] mainly demonstrates type checking but not type inference. There has been several approaches to automate type checker generation afterwards (e.g., TCG [9], Typical [19], TyCC in the TyS Framework [18]).

HMX [17] is a representative example of an LP based executable specification of polymorphic type systems, later on implemented using CHRs [3]. HMX is designed to support the type system with polymorphic type inference and its extensions. Tomb and Flanagan [22] specified monomorphic subtyping

rules in Prolog to automate type inference by partial evaluation and demonstrate that their implementation can be used for sign analysis of Java programs. Recently there have been approaches using non-standard LP semantics involving coinductive flavours [4, 8] and early work attempting both specification and formal reasoning of type systems [11].

There is a recent trend of utilizing off-the-shelf logic solvers to implement advanced features or extensions of type systems. For instance, Vazou and others [23] used SMT solvers to implement Liquid Haskell, which is an extension of Haskell with refinement types, Hashimoto and Unno [12] implemented a refinement type inference algorithm using an SMT solver.

## 7 Summary and future work

We have developed a series of type system specifications including type polymorphism, type-constructor polymorphism, kind polymorphism (not shown in this paper), and extensible records based on row polymorphism, which is the main contribution of this paper. Further examples including kind polymorphism are available online on the TIPER homepage <http://kyagrd.github.io/tiper/>.

In contrast to the approach of having two distinct phases for constraint generation and solving in prior related work (e.g., [17, 22]), the execution of our specification can be viewed as interleaving constraint generation and solving. The typing constraints are generated by recursive invocations of the typing rules, following the recursive structure of the term syntax. In addition, the kinding constraints are generated in the midst of resolving typing constraints. But there is a separation of stages of type inference and kind inference, where the unification constraints of the latter stage are being generated from the former stage.

The features we would like to specify next are *first-class polymorphism* [20], of course with annotations because complete inference for System F or  $F_\omega$  is known to be impossible, and *term-indexed types* [1], which leads to more than two-stages of rule execution.

Our specifications have educational value but lack informative error messages, which is important for practical implementations. For future work, we would like to search for solutions on error reporting and better ways of handling multi-staged execution of rules with different sets of logic variables for each stage.

## References

- [1] Ki Yung Ahn, Tim Sheard, Marcelo Fiore & Andrew M. Pitts (2013): *System Fi: a Higher-Order Polymorphic Lambda Calculus with Erasable Term-Indices*, pp. 15–30. Springer Berlin Heidelberg, doi:[10.1007/978-3-642-38946-7\\_4](https://doi.org/10.1007/978-3-642-38946-7_4).
- [2] Ki Yung Ahn & Andrea Vezzosi (2016): *Functional and Logic Programming: 13th International Symposium, FLOPS 2016*, chapter Executable Relational Specifications of Polymorphic Type Systems Using Prolog, pp. 109–125. LNCS, Springer International Publishing, Cham, doi:[10.1007/978-3-319-29604-3\\_8](https://doi.org/10.1007/978-3-319-29604-3_8).
- [3] Sandra Alves & Mrio Florido (2002): *Type Inference using Constraint Handling Rules*. *Electronic Notes in Theoretical Computer Science* 64, pp. 56 – 72, doi:[10.1016/S1571-0661\(04\)80346-3](https://doi.org/10.1016/S1571-0661(04)80346-3). WFLP 2001, International Workshop on Functional and (Constraint) Logic Programming, Selected Papers.
- [4] Davide Ancona, Giovanni Lagorio & Elena Zucca (2008): *Type Inference by Coinductive Logic Programming*. In: *TYPES 2008, LNCS 5497*, Springer, pp. 1–18.
- [5] Gilad Bracha (2014): *Whither Web programming?* Keynote talk at QCon New York. Available at <https://qconnewyork.com/ny2014/keynote/whither-web-programming.html>.
- [6] Thierry Despeyroux (1984): *Executable Specification of Static Semantics*. In: *Semantics of Data Types, LNCS 173*, Springer, pp. 215–233.

- [7] Facebook (2014): *Flow | a static type checker for javascript*. <http://flowtype.org/>. [Online: accessed 2016-05-05, first released in 2014].
- [8] Peng Fu, Ekaterina Komendantskaya, Tom Schrijvers & Andrew Pond (2016): *Functional and Logic Programming: 13th International Symposium, FLOPS 2016*, chapter Proof Relevant Corecursive Resolution, pp. 126–143. Springer International Publishing, Cham.
- [9] Holger Gast (2004): *A generator for type checkers*. Ph.D. thesis, University of Tübingen, Germany.
- [10] Benedict R. Gaster & Mark P. Jones (1996): *A Polymorphic Type System for Extensible Records and Variants*. Technical Report NOTTCS-TR-96-3, Department of Computer Science, University of Nottingham.
- [11] Sylvia Grewe, Sebastian Erdweg, Pascal Wittmann & Mira Mezini (2015): *Type systems for the masses: deriving soundness proofs and efficient checkers*. In: *ACM SPLASH Onward!*, pp. 137–150.
- [12] Kodai Hashimoto & Hiroshi Unno (2015): *Refinement Type Inference via Horn Constraint Optimization*, pp. 199–216. Springer Berlin Heidelberg, doi:[10.1007/978-3-662-48288-9\\_12](https://doi.org/10.1007/978-3-662-48288-9_12).
- [13] Mark P. Jones (1992): *A Theory of Qualified Types*. In Bernd Krieg-Brückner, editor: *ESOP, Lecture Notes in Computer Science 582*, Springer, pp. 287–306. Available at [http://dx.doi.org/10.1007/3-540-55253-7\\_17](http://dx.doi.org/10.1007/3-540-55253-7_17).
- [14] Ekaterina Komendantskaya & Frantisek Farka (2016): *CoALP-Ty'16*. CoRR abs/1612.03032. Available at <http://arxiv.org/abs/1612.03032>.
- [15] André Murbach Maidl, Fabio Mascarenhas & Roberto Ierusalimsky (2014): *Typed Lua: An Optional Type System for Lua*. In: *Proceedings of the Workshop on Dynamic Languages and Applications, Dyla 2014*, pp. 3:1–3:10, doi:[10.1145/2617548.2617553](https://doi.org/10.1145/2617548.2617553).
- [16] Microsoft (2012): *TypeScript - JavaScript that scales*. <http://www.typescriptlang.org/>.
- [17] Martin Odersky, Martin Sulzmann & Martin Wehr (1999): *Type Inference with Constrained Types*. *Theor. Pract. Object Syst.* 5(1), pp. 35–55.
- [18] Francisco Ortin, Daniel Zapico Palacio, Jose Quiroga & Miguel García (2014): *TyS - A Framework to Facilitate the Implementation of Object-Oriented Type Checkers*. In: *SEKE*, Knowledge Systems Institute Graduate School, pp. 150–155.
- [19] Anh Le Robert Grimm, Laune Harris (2007): *Typical: Taking the Tedium Out of Typing*. Technical Report TR2007-904, New York University.
- [20] Claudio V. Russo & Dimitrios Vytiniotis (2009): *QML: Explicit First-class Polymorphism for ML*. In: *Proceedings of the 2009 ACM SIGPLAN Workshop on ML, ML '09*, ACM, New York, NY, USA, pp. 3–14, doi:[10.1145/1596627.1596630](https://doi.org/10.1145/1596627.1596630).
- [21] Frieder Stolzenburg (1996): *Membership-Constraints and Complexity in Logic Programming with Sets*. In: *Frontiers in Combining Systems*, Kluwer Academic, pp. 285–302.
- [22] Aaron Tomb & Cormac Flanagan (2005): *Automatic Type Inference via Partial Evaluation*. In: *Proceedings of the 7th ACM SIGPLAN International Conference on Principles and Practice of Declarative Programming, PPDP '05*, ACM, pp. 106–116, doi:[10.1145/1069774.1069784](https://doi.org/10.1145/1069774.1069784).
- [23] Niki Vazou, Eric L. Seidel, Ranjit Jhala, Dimitrios Vytiniotis & Simon Peyton-Jones (2014): *Refinement Types for Haskell*. In: *Proceedings of the 19th ACM SIGPLAN International Conference on Functional Programming, ICFP '14*, ACM, New York, NY, USA, pp. 269–282.
- [24] Dimitrios Vytiniotis, Simon Peyton Jones, Tom Schrijvers & Martin Sulzmann (2011): *Outsidein(x) Modular Type Inference with Local Assumptions*. *J. Funct. Program.* 21(4-5), pp. 333–412.
- [25] Brent A. Yorgey, Stephanie Weirich, Julien Cretin, Simon Peyton Jones, Dimitrios Vytiniotis & José Pedro Magalhães (2012): *Giving Haskell a Promotion*. In: *Proceedings of the 8th ACM SIGPLAN Workshop on Types in Language Design and Implementation, TLDI '12*, ACM, pp. 53–66, doi:[10.1145/2103786.2103795](https://doi.org/10.1145/2103786.2103795).